United Kingdom Hydrographic Office

HM NAUTICAL ALMANAC OFFICE

NAO TECHNICAL NOTE

No. 63

The Computation of Angular Atmospheric Refraction at
Large Zenith Angles

by

C.Y. Hohenkerk and *A.T. Sinclair*



HM Nautical Almanac Office
UK Hydrographic Office
Admiralty Road, Taunton
TA1 2DN England

Telephone: +44 (0) 1823 337900
Facsimile: +44 (0) 1823 335396
e-mail: hmnao@ukho.gov.uk
WWW http://www.hmnao.com/

Particle Physics and Astronomy Research Council

HM Nautical Almanac Office

Royal Greenwich Observatory

NAO TECHNICAL NOTE

No. 63 1985 April

The computation of angular atmospheric refraction at large zenith angles

by

C. Y. Hohenkerk and *A.T. Sinclair*

Summary

The bending of a light ray as it passes through the atmosphere is calculated by numerical evaluation of the refraction integral, expressed in a form that avoids numerical difficulties at a zenith angle of 90°. A polytropic model atmosphere is used, and the parameters of the model are varied in order to determine the effect on the computed refraction value. Good agreement is obtained with other evaluations of refraction for zenith angles up to 75°. Beyond that it becomes sensitive to the values of the parameters used, particularly the pressure, temperature, temperature lapse rate, and wavelength of the light ray. Good agreement is obtained with other evaluations provided the same parameter values are used.

Royal Greenwich Observatory,
Madingley Road
Cambridge
CB3 0EZ England

Telephone: +44 (0) 1223 374000
Facsimile: +44 (0) 1223 374700
Email: nao@ast.cam.ac.uk
WWW: http://www.ast.cam.ac.uk/~nao/

The Computation of Angular Atmospheric Refraction at Large Zenith Angles.

by

C. Y. Hohenkerk and *A. T. Sinclair*

1. Introduction.

As a ray of light passes through the atmosphere its direction is altered by the effects of atmospheric refraction. The amount of refraction depends on the variation of the refractive index of air along the path of the ray. Models of the atmosphere can be formulated from which the refractive index at any height can be calculated. The equations describing the bending of the light ray can be solved by analytical techniques, to give expressions for the refraction effect. For zenith angles up to about 75°, Saastamoinen (1972) gives a compact expression for the refraction, with explicit dependence on the pressure, temperature and water vapour pressure at the observing site. Correction tables are given for the height of the observing site above sea level, and for the wavelength of the light ray. For zenith angles greater than 75° Garfinkel's analytical theory (1944, 1967) can be used. This takes the form of a rather lengthy computer program, and is not particularly simple to use. Auer and Standish (1979) describe a method of evaluating refraction by numerical quadrature. This leads to a computer program that is quick and simple to use, and in which the atmospheric model used can be varied easily.

In this note we describe an implementation of the method recommended by Auer and Standish. The atmospheric model used is fairly standard and is very similar to that used by Saastamoinen. The pressure, temperature, humidity, height, latitude and wavelength of light are parameters to be input to the calculation. In addition the parameters of the atmospheric model can be easily varied. For zenith angles from 0° to 75° the computed value of refraction is insensitive to the parameters of the model, and is in good agreement with values from other sources (eg. Saastamoinen). For angles greater than 75° the value computed is sensitive to the parameters of the model, particularly the temperature lapse rate. If the same values are chosen for the critical parameters then good agreement is obtained with Garfinkel's analytical method. The calculations can be done to any specified precision; usually $0\rlap{.}''001$ was used in the results described in this note. The real accuracy of the results, is of course, considerably worse, particularly at large zenith angles. Uncertainties in the assumed values of the parameters of the model could lead to errors in the calculated refraction of up to $20''$ at a zenith angle of 90°. In addition, local atmospheric effects which are not modelled can be important.

In practice, the use of the method described in this note for calculating refraction at zenith angles near 90° should be regarded as a source of a well defined reference value, which allows for the principal changes of refraction due to variations of pressure, temperature and humidity, and perhaps temperature lapse rate also if this could be measured. Thereafter observations would be needed, and the residuals of the observations from the reference values could perhaps be used to model local variations of refraction.

2. Refraction formulae.

The formulae describing the bending of a light ray through the atmosphere are given in many text books, for example, Smart (1949). A simple description is given in Sinclair (1982). In this note we simply quote the standard formulae.



Figure 1. Path of a light ray through the atmosphere.

Figure 1 shows the path of a light ray through the atmosphere, to an observer at O, at distance r_0 from the centre of the Earth. At a general point P along the path the zenith angle is z and the radial distance is r. By Snell's law of refraction, the amount of bending of the ray relative to a **fixed** direction as the ray passes though the zone $r + dr$ to r, in which the refractive index changes from $n + dn$ to dn, is $\tan z \, dn/n$. Hence the total bending of the light ray is

$$\xi = \int_1^{n_0} \frac{\tan z}{n} \, dn = \int_\infty^{r_0} \frac{\tan z}{n} \frac{dn}{dr} \, dr \tag{1}$$

where n_0 is the refractive index at the observer, and the refractive index outside the atmosphere is 1. Also, from Snell's law applied to a spherically symmetric atmosphere, we have

$$n \, r \sin z = n_0 \, r_0 \sin z_0 \tag{2}$$

where z_0 is the value of z at the observer.

So, provided the refractive index n and dn/dr are known at any radial distance r, the value of z can be calculated from equation (2), and so the integrand of equation (1) can be evaluated. The evaluation of the integrand by analytical methods is complicated, particularly at large zenith angles. However the numerical evaluation by quadrature is reasonably straightforward, except for zenith angles near 90° when the integrand becomes very large.

Auer and Standish (1979) recommended a change to z as the independent variable, in order to prevent the integrand becoming infinite. This also makes the integrand a more slowly varying function over the whole range of values of z used, and so simplifies the numerical quadrature. We obtain

$$\xi = \int_0^{z_0} \frac{\tan z}{n} \, \frac{dn}{d(nr)} \, \frac{d(nr)}{dz} \, dz$$

From equation (2), $n\,r$ is a simple function of z, giving

$$\frac{d(nr)}{dz} = -\, n\,r \cot z$$

And,

$$\frac{dn}{d(nr)} = \frac{dn}{dz} \Big/ \frac{d(nr)}{dz} = \frac{dn/dr}{n + r\,dn/dr}$$

Hence,

$$\xi = -\int_0^{z_0} \frac{r\,dn/dr}{n + r\,dn/dr} \, dz \tag{3}$$

This integral is well-behaved near $z = 90°$, and is evaluated by quadrature. In practice the integral is not evaluated to $z = 0$. The lower extreme of z is taken to be a value corresponding to a height of about 80 km, at which point refraction is negligible.

The quadrature takes equal steps in z, and at each value of z it is necessary to calculate the corresponding value of r, and then of n and dn/dr. This is done by solving equation (2) for r. It is required to find the root of $F(r) = 0$, where

$$F(r) = n\,r - \frac{n_0\,r_0 \sin z_0}{\sin z}$$

where n is a known function of r, and the values of n_0, r_0, z_0 and z are all known. The root is found by Newton-Raphson iteration :

$$r_{i+1} = r_i - \frac{F(r_i)}{F'(r_i)}$$

Convergence is rapid, so that the initial estimate of r is not critical; the value from the previous quadrature step is adequate.

Solution of this equation will fail for a vertical ray, as $z = 0$. However refraction at $z = 0$ is zero anyway, so there is no need to compute it.

3. The model of the atmosphere.

The same model of the atmosphere is used as in Sinclair (1982), so only the resulting formulae are quoted in this note. The physical assumptions made in this model are

1. The temperature decreases at a constant rate within the troposphere, up to the tropopause at about 11 km height. Above the tropopause (in the stratosphere) the temperature remains constant.

2. The atmosphere obeys the perfect gas law, for the combined mixture of dry air and water vapour, and also for the dry air and water vapour separately.

3. The atmosphere is in hydrostatic equilibrium.

4. The relative humidity remains constant through the troposphere, equal to its value at the observer.

The following parameters are needed to describe the variation of temperature and pressure along the path of the ray:

z_0	the observed zenith angle of the light ray
h	height of observer above the geoid (m)
ϕ	latitude of observer
h_t	height of tropopause above geoid(m) ($\approx 11\,000$ m)
h_s	height above geoid(m) at which refraction is negligible (say $80\,000$ m)
P_0	total atmospheric pressure at observer (mb)
P_{w0}	partial pressure of water vapour at observer (mb), if relative humidity is known, then $P_{w0} = R_h\,(T_0/247{\cdot}1)^{18{\cdot}36}$
T_0	temperature at observer (°K)
α	temperature lapse rate $^\circ\mathrm{K\,m^{-1}}$ ($\approx 0{\cdot}0065$)
δ	exponent of temperature dependence of water vapour pressure. The likely value is in the range 18-21. We adopt the value 18·36, but there is no necessity for the value to be the same as the exponent used above in the humidity expression)
λ	wavelength of light in microns (conventional value for mean starlight is $0{\cdot}574\,\mu$m)

Constants needed are

$$R = 8314{\cdot}36 \qquad \text{(universal gas constant)}$$
$$M_d = 28{\cdot}966 \qquad \text{(mol. wt. of dry air)}$$
$$M_w = 18{\cdot}016 \qquad \text{(mol. wt. of water vapour)}$$
$$\overline{g} = 9{\cdot}784\,(1 - 0{\cdot}0026\cos 2\phi - 0{\cdot}0000\,0028\,h)$$

Then, calculate

$$r_0 = 6\,378\,120 + h$$
$$r_t = 6\,378\,120 + h_t$$
$$r_s = 6\,378\,120 + h_s$$
$$\gamma = \overline{g}\,M_d/(R\,\alpha)$$
$$A = \left(287{\cdot}604 + \frac{1{\cdot}6288}{\lambda^2} + \frac{0{\cdot}0136}{\lambda^4}\right)\frac{273{\cdot}15}{1013{\cdot}25}$$

Then, in the troposphere (ie. for $r_0 < r < r_t$) the total pressure, water vapour pressure and temperature, and hence n and dn/dr, are calculated from

$$T = T_0 - \alpha\,(r - r_0)$$
$$P_w = P_{w0}\,(T/T_0)^{\delta}$$
$$P = \left[P_0 + \left(1 - \frac{M_w}{M_d}\right)\frac{\gamma}{\delta-\gamma}\,P_{w0}\right]\left(\frac{T}{T_0}\right)^{\gamma} - \left(1 - \frac{M_w}{M_d}\right)\frac{\gamma}{\delta-\gamma}\,P_w$$
$$n = 1 + 10^{-6}(A\,P - 11{\cdot}2684\,P_w)/T$$
$$\frac{dn}{dr} = -\frac{(\gamma-1)\,\alpha\,A\,10^{-6}}{T_0^2}\left[P_0 + \left(1 - \frac{M_w}{M_d}\right)\frac{\gamma}{\delta-\gamma}\,P_{w0}\right]\left(\frac{T}{T_0}\right)^{\gamma-2} + \frac{(\delta-1)\,\alpha\,10^{-6}}{T_0^2}\left[A\left(1 - \frac{M_w}{M_d}\right)\frac{\gamma}{\delta-\gamma} + 11{\cdot}2684\right]P_{w0}\left(\frac{T}{T_0}\right)^{\delta-2}$$

Let the values of T, P and n at the tropopause ($r = r_t$) be T_t, P_t, n_t.

In the stratosphere ($r > r_t$), the corresponding expressions are:

$$T = T_t \qquad \text{constant}$$
$$P_w = 0$$
$$P = P_t \exp\left[-\frac{\overline{g}\,M_d}{R\,T_t}\,(r - r_t)\right]$$
$$\text{Hence} \quad n = 1 + (n_t - 1)\exp\left[-\frac{\overline{g}\,M_d}{R\,T_t}\,(r - r_t)\right]$$
$$\frac{dn}{dr} = -\frac{\overline{g}\,M_d}{R\,T_t}\,(n_t - 1)\exp\left[-\frac{\overline{g}\,M_d}{R\,T_t}\,(r - r_t)\right]$$

4. Numerical evaluation of the refraction integral.

The model of the atmosphere used has a discontinuity in the temperature gradient at the tropopause, and this causes a discontinuity in dn/dr, and hence in the integrand of equation (3) also. This makes it necessary to evaluate the integral by separate quadratures in the troposphere and the stratosphere, using the appropriate value of dn/dr at the tropopause in each case. Hence the integral in equation (3), from $z = 0$ to z_0, is in practice evaluated in two parts, $z = z_s$ to z_t in the stratosphere, and from $z = z_t$ to z_0 in the troposphere, where

$$n_0 = 1 + 10^{-6}(A\,P_0 - 11{\cdot}2684\,P_{w0})/T_0$$
$$z_t = \sin^{-1}\left(\frac{n_0\;r_0 \sin z_0}{n_t\,r_t}\right)$$
$$z_s = \sin^{-1}\left(\frac{n_0\;r_0 \sin z_0}{n_s\,r_s}\right)$$

and the distance r_s (at height h_s) is such that refraction is negligible; ie $n_s = 1$

The two parts of the integral were evaluated by Simpson's rule, and the use of z as the independent variable makes it efficient to use a constant step in z throughout the range of each part of the integral. The step length needed to achieve the specified precision was determined by repeatedly halving it, until successive evaluations agreed. For a precision of $0\overset{\prime\prime}{.}01$ it was found that 32 steps were needed in the troposphere, and 128 in the stratosphere, at 60° zenith angle.

The appendix to this paper gives the listing of FORTRAN subroutines for the evaluation of the refraction integral. A call to the subroutine AREF, with input arguments the observed zenith angle, height of observer, temperature, pressure, relative humidity, wavelength of the light ray, observer's latitude, temperature lapse rate and the accuracy required, will return the amount of refraction. The units and variable types of the parameters are specified in the heading of the subroutine. The subroutine calls two other subroutines, ATMOSTRO and ATMOSSTR.

As explained in Section 6, the refraction at large zenith angles has a strong dependence on the temperature lapse rate α, and so this is included as an argument of the subroutine. If a local value is not known then a value of $0{\cdot}0065°\mathrm{K\,m^{-1}}$ should be used (the sign of the value does not matter as the absolute value is used).

The calculated values of refraction were found to be insensitive to the other parameters of the model, and so the following values are incorporated into the subroutine (specified in the DATA statement):

$\delta = 18{\cdot}36$	(water vapour exponent)
$h_t = 11\,000$	(height of tropopause)
$h_s = 80\,000$	(height at which refraction is negligible)

These values were used for the comparisons in Section 5.

5. Comparisons with other sources.

Table 1 gives a comparison of the values of refraction computed by the methods described in this note with values from Saastamoinen's formula, and tabulations in the *Star Almanac*. Saastamoinen's formula (1972 b) is

$$\xi = 16''271\, Q \tan z_0 (1 + 0{\cdot}0000\,394\, Q \tan^2 z_0) - 0''0000\,749\, P_0 (\tan z_0 + \tan^3 z_0)$$
$$\text{where} \quad Q = (P_0 - 0{\cdot}156\, P_{w0})/T_0,$$

and the other symbols are as used in this note. The formula is for a wavelength of $0{\cdot}574\,\mu$m, and for an observer at sea level. Saastamoinen gives correction tables for other wavelengths and for height above sea level. The formulae is intended for $z_0 \leq 70°$, but additional corrections to allow for the curvature of the atmosphere are given for z_0 from 66° to 80°. These additional corrections range from 0″02 to 1″14, and they are included in the comparison below. His formula incorporates a temperature gradient of $0{\cdot}0065°\mathrm{K\,m^{-1}}$, and the same value is used in our program for the comparison.

Table 1. Comparison with Saastamoinen and *Star Almanac*

Pressure at observer	:	1005 mb
Temperature at observer	:	280·15°K
Humidity at observer	:	80%
Wavelength of ray	:	$0{\cdot}574\,\mu$m
Lapse rate, α	:	$0{\cdot}0065°\mathrm{K\,m^{-1}}$
Height of observer	:	sea level
Latitude of observer	:	50°

Observed zenith angle	*Star Almanac*	Refraction Saastamoinen	this note
°	″	″	″
10	10	10·27	10·27
20	21	21·19	21·19
30	34	33·60	33·61
40	49	48·81	48·83
45	58	58·16	58·17
50	69	69·27	69·29
55	83	82·95	82·98
60	100	100·50	100·53
65	124	124·20	124·25
70	159	158·61	158·66
72	177	177·31	177·35
74	200	200·32	200·38
76	229	229·42	229·48
78	267	267·41	267·48
80	319	319·10	319·18

The *Star Almanac for Land Surveyors* (HMSO) gives a table of refraction for z_0 up to 80°, based on Harzer's tables (1924). The table is for a pressure of 1005 mb and a temperature of 7°C, with correction factors for other temperatures and pressures. Hence these values are used in the comparison. The wavelength and humidity are not specified for the *Star Almanac* table, so $\lambda = 0{\cdot}574\,\mu$m is used (a conventional value for mean starlight), and a humidity of 80% is assumed.

Table 2 gives a comparison with values obtained from Garfinkel's program, based on his theory (1944, 1967). The *Nautical Almanac* (HMSO) contains a table of refraction at high zenith angles, which was computed from Garfinkel's 1944 theory, and so the same constants were used in Garfinkel's program and in the program described in this note. The *Nautical Almanac* data agree to the precision quoted (0′1) with the data from

Table 2. Comparison with Garfinkel's program (as used in *Nautical Almanac*)

Pressure at observer	:	1010 mb
Temperature at observer	:	283·15°K
Humidity at observer	:	0%
Wavelength of ray	:	0·50169 μm
Lapse rate, α	:	see tabulation
Height of observer	:	sea level
Latitude of observer	:	50°

Observed zenith angle	Garfinkel $\alpha = 0{\cdot}0056\,94$	Refraction this note $\alpha = 0{\cdot}0056\,94$	this note $\alpha = 0{\cdot}0065$
°	″	″	″
75	214·33	214·20	214·20
76	229·81	229·66	229·66
77	247·48	247·32	247·32
78	267·85	267·68	267·68
79	291·59	291·41	291·40
80	319·60	319·40	319·39
81	353·12	352·91	352·88
82	393·92	393·68	393·63
83	444·53	444·25	444·17
84	508·77	508·46	508·30
85	592·59	592·21	591·92
86	705·58	705·12	704·52
87	864·05	863·44	862·10
88	1097·13	1096·26	1093·02
89	1460·31	1458·93	1450·38
90	2068·28	2065·77	2041·04

Garfinkel's program given in Table 2, showing that Garfinkel's theory and program are in agreement. The pressure and temperature used in the *Nautical Almanac* table are 1010 mb and 10°C. Garfinkel's theory does not include any modelling of water vapour, so it was assumed that this corresponds to zero humidity. The theory adopts a refractive index of air of 1·0002 9429 at 760 mm pressure and 273·15°K temperature, and this corresponds to a wavelength of the ray of 0·5016 9 μm in the model used in this note. The value of the temperature lapse rate used is $\alpha = 0{\cdot}0056\,94$°K m^{-1}, and so the same value is used in the program described in this note, but for comparison a more modern standard value of 0·0065°K m^{-1} is used also.

6. Sensitivity of refraction to parameters of model.

The model of the atmosphere described in Section 3 contains several physical parameters, whose values are slightly variable, and are not likely to be known with great accuracy. Hence the refraction was calculated at zenith angles of $z_0 = 60°$ and 90° when these parameters were varied in turn by plausible amounts from the standard values adopted in this note. The same was also done for the parameters measured at the observer (pressure, temperature etc) in order to determine the accuracy with which these parameters must be known. Standard values of $P_0 = 1013$ mb and $T_0 = 273{\cdot}15$°K were used, and variations were made from these values. The variation with height of observer is somewhat artificial since it ignores the much larger variation that occurs due to the consequent drop in pressure. The results are given in Table 3, (the signs of the changes in refraction correspond to increases of the values of the parameters by the amounts indicated). It can be seen that refraction is insensitive to the values of h_t, h_s, and δ, and so fixed values are used in the subroutine. All other parameters are included as arguments of the subroutine. The determination of most of these to sufficient accuracy will present no problem; the only difficulty will be to determine the temperature lapse rate α. The effect of this is considerable at large zenith angles, as can also be seen from Table 2.

Table 3. Sensitivity of refraction to various parameters at zenith angles 60° and 90°

Parameter	Amount changed	Refraction At $z_0 = 60°$	At $z_0 = 90°$
		$''$	$''$
Wavelength	$0{\cdot}1\,\mu$m(1000Å)	$-\ 1$	-17
Pressure	10 mb	$+\ 1$	$+23$
Temperature	$2{\cdot}5$°K	$-\ 1$	-34
Humidity	100%	0	$-\ 5$
α (lapse rate)	$0{\cdot}0005$°K m^{-1}	0	-17
h_t (ht tropopause)	2000 m	0	$-\ 2$
h_s (upper limit)	20 000 m	0	0
δ (water vapour index)	2	0	$-\ 1$
latitude	45°	0	$+\ 4$
h_0 (ht of observer)	5 000 m	0	$+\ 6$

7. Conclusions.

It has been shown that the bending of a light ray due to refraction in the atmosphere can be calculated fairly simply by numerical evaluation of an integral. A computer subroutine to do this is given, and comparisons with several analytical evaluations of refraction show good agreement. With the numerical method described in this note it is simple to change the values of the parameters that describe the atmospheric model used. The value of refraction is fairly insensitive to most of these parameters, but at large zenith angles (say greater than 75°), the values of pressure, temperature, temperature lapse rate and the wavelength of the light ray are important. For a precision of $1''$ at a zenith angle of 90° these quantities must be measured to accuracies of $0{\cdot}4$ mb, $0{\cdot}1$°K, $0{\cdot}006\,\mu$m and $0{\cdot}00003$°K m^{-1} respectively. The real accuracy of the calculated value of refraction will be considerably worse that $1''$, due to unmodelled local effects which become important at large zenith angles. The precise calculated value provides a reference relative to which local variations can be studied.

References

Auer, L.H. and Standish, E.M. (1979). Astronomical refraction: Computational method for all zenith angles. Yale University Astronomy Department note.

Garfinkel, B. (1944). An investigation in the theory of astronomical refraction. Astr. J. **50**, 169-179.

Garfinkel, B. (1967). Astronomical refraction in a polytropic atmosphere. Astr. J. **72**, 235-254.

Harzer, P. (1924). Gebrauchstabellen zur Berechnung der Ablenkungen der Lichtstrahlen in der Atmosphäre der Erde für die Beobachtungen am großen Kieler Meridiankreise. Publikation der Sternwarte in Kiel. **14**.

Saastamoinen, J. (1972a). Contributions to the theory of atmospheric refraction. Bulletin Géodésique. 279-298.

Saastamoinen, J. (1972b). Introduction to the practical computation of astronomical refraction. Bulletin Géodésique. 383-397.

Smart, W.M. (1949). Text book on Spherical Astronomy. Cambridge University Press.

Sinclair, A.T. (1982). The effect of atmospheric refraction on laser ranging data. NAO Technical Note no. 59. Royal Greenwich Observatory.

```
      SUBROUTINE AREF(Z0,H0,T0,P0,UPS,WL,PH,AS,EPS,REF)
C
C A subroutine to calculate bending refraction.      The method is based on
C N.A.O Technical Notes 59 and 63 and a paper by Standish and Auer
C 'Astronomical Refraction:  Computational Method for all Zenith Angles'.
C
C Variable Type Function
C
C Input:-
C
C  Z0      D    The observed zenith distance of the object in degrees
C  H0      D    The height of the observer above sea level in metres
C  T0      D    The temperature at the observer in degrees K
C  P0      D    The pressure at the observer in millibars
C  UPS     D    The relative humidity at the observer
C  WL      D    The wavelength of the light at the observer in micrometres
C  PH      D    The latitude of the observer in degrees
C  AS      D    The temperature lapse rate in degrees K/metre in the
C               troposphere, the absolute value is used
C  EPS     D    The precision required in seconds of arc
C
C Output:-
C
C  REF     D    The refraction at the observer in degrees
C
C Vax 11/750   1984 September         Catherine Hohenkerk      HMNAO
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DOUBLE PRECISION MD,MW,N,N0,NT,NTS,NS,A(10)
C
      DATA GCR/8314.36D0/,MD/28.966D0/,MW/18.016D0/,S/6378120.0D0/,
     * GAMMA/18.36D0/,HT/11000.0D0/,HS/80000.0D0/,DGR/0.01745329252D0/,
     * Z2/11.2684D-06/
C
C The refraction integrand
C
      REFI(R,N,DNDR) = R*DNDR/(N + R*DNDR)
C
C Set up parameters defined at the observer for the atmosphere
C
      GB= 9.784D0*(1.0D0 - 0.0026D0*COS(2.0D0*PH*DGR) - 0.00000028D0*H0)
      Z1 = (287.604D0 + 1.6288D0/(WL*WL) + 0.0136D0/(WL*WL*WL*WL))
     *          *(273.15D0/1013.25D0)*1.0D-06
      A(1) = ABS(AS)
      A(2) = (GB*MD)/GCR
      A(3) = A(2)/A(1)
      A(4) = GAMMA
      PW0 = UPS*(T0/247.1D0)**A(4)
      A(5) = PW0*(1.0D0 - MW/MD)*A(3)/(A(4)-A(3))
      A(6) = P0 + A(5)
      A(7) = Z1*A(6)/T0
      A(8) = ( Z1*A(5) + Z2*PW0)/T0
      A(9) = (A(3) - 1.0D0)*A(1)*A(7)/T0
      A(10) = (A(4) - 1.0D0)*A(1)*A(8)/T0
C
C At the Observer
C
      R0 = S + H0
      CALL ATMOSTRO(R0,T0,A,R0,T00,N0,DNDR0)
      SK0 = N0 * R0 * SIN(Z0*DGR)
```

```
      F0 = REFI(R0,N0,DNDR0)
C
C At the Tropopause in the Troposphere
C
      RT = S + HT
      CALL ATMOSTRO(R0,T0,A,RT,TT,NT,DNDRT)
      ZT = ASIN(SK0/(RT*NT))/DGR
      FT = REFI(RT,NT,DNDRT)
C
C At the Tropopause in the Stratosphere
C
      CALL ATMOSSTR(RT,TT,NT,A(2),RT,NTS,DNDRTS)
      ZTS = ASIN(SK0/(RT*NTS))/DGR
      FTS = REFI(RT,NTS,DNDRTS)
C
C At the stratosphere limit
C
      RS = S + HS
      CALL ATMOSSTR(RT,TT,NT,A(2),RS,NS,DNDRS)
      ZS = ASIN(SK0/(RS*NS))/DGR
      FS = REFI(RS,NS,DNDRS)
C
C Integrate the refraction integral in the troposhere and stratosphere
C ie  Ref  = Ref troposhere + Ref stratopshere
C
C Initial step lengths etc
C
      REFO = -999.999
      IS = 16
      DO K = 1,2
       ISTART = 0
       FE = 0.0D0
       FO = 0.0D0
C
       IF (K.EQ.1) THEN
        H = (ZT - Z0)/DBLE(IS)
        FB = F0
        FF = FT
        ELSE IF (K.EQ.2) THEN
         H = (ZS - ZTS  )/DBLE(IS)
         FB = FTS
         FF = FS
       ENDIF
C
       IN = IS - 1
       IS = IS/2
       STEP = H
 200   CONTINUE
C
       DO I = 1,IN
C
        IF (I.EQ.1.AND.K.EQ.1) THEN
         Z = Z0 + H
         R = R0
         ELSE IF (I.EQ.1.AND.K.EQ.2) THEN
          Z = ZTS + H
          R = RT
          ELSE
           Z = Z + STEP
        ENDIF
C
```

```
C
C Given the Zenith distance (Z) find R
C
        RG = R
        DO J = 1,4
         IF (K.EQ.1) THEN
          CALL ATMOSTRO(R0,T0,A,RG,TG,N,DNDR)
          ELSE IF (K.EQ.2) THEN
           CALL ATMOSSTR(RT,TT,NT,A(2),RG,N,DNDR)
         ENDIF
         RG = RG - ( (RG*N - SK0/SIN(Z*DGR))/(N + RG*DNDR) )
        ENDDO
        R = RG
C
C Find Refractive index and Integrand at R
C
        IF (K.EQ.1) THEN
         CALL ATMOSTRO(R0,T0,A,R,T,N,DNDR)
         ELSE IF (K.EQ.2) THEN
          CALL ATMOSSTR(RT,TT,NT,A(2),R,N,DNDR)
        ENDIF
C
        F = REFI(R,N,DNDR)
C
        IF (ISTART.EQ.0.AND.MOD(I,2).EQ.0) THEN
         FE = FE + F
         ELSE
          FO = FO + F
        ENDIF
       ENDDO
C
C Evaluate the integrand using Simpson's Rule
C
       REFP = H*(FB + 4.0D0*FO + 2.0D0*FE + FF)/3.0D0
C
       IF (ABS(REFP-REFO).GT.0.5D0*EPS/3600.0D0) THEN
        IS = 2*IS
        IN = IS
        STEP = H
        H = H/2.0D0
        FE = FE + FO
        FO = 0.0D0
        REFO = REFP
        IF (ISTART.EQ.0) ISTART = 1
        GO TO 200
       ENDIF
       IF (K.EQ.1)    REFT = REFP
      ENDDO
C
      REF = REFT + REFP
C
      END
```

```
      SUBROUTINE ATMOSTRO(R0,T0,A,R,T,N,DNDR)
C
C Atmosphere of the troposphere
C
C Variable Type Function
C
C Input:-
C
C  R0       D    The height of the observer from the centre of the Earth
C  T0       D    The temperature at the observer in degres K
C  A       D(10) Constants defined at the observer
C  R        D    The current distance from the centre of the Earth,metres
C
C Output:-
C
C  T        D    The temperature at R in degrees K
C  N        D    The refractive index at R
C  DNDR     D    The rate the refractive index is changing at R
C
C Vax 11/750   1984 August               Catherine Hohenkerk
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DOUBLE PRECISION N,A(10)
C
      T = T0 - A(1)*(R-R0)
      TT0 = T/T0
      TT01 = TT0**(A(3)-2.0D0)
      TT02 = TT0**(A(4)-2.0D0)
      N = 1.0D0 + ( A(7)*TT01 - A(8)*TT02 )*TT0
      DNDR = -A(9)*TT01 + A(10)*TT02
C
      END

      SUBROUTINE ATMOSSTR(RT,TT,NT,A,R,N,DNDR)
C
C Atmosphere of the stratosphere
C
C Variable Type Function
C
C Input:-
C
C  RT       D    The height of the tropopause from the centre of the
C                 Earth in metres
C  TT       D    The temperature at the tropopause in degrees K
C  NT       D    The refractive index at the tropopause
C  A        D    Constant of the atmospheric model = G*MD/R
C  R        D    The current distance from the centre of the Earth in metres
C
C Output:-
C
C  N        D    The refractive index at R
C  DNDR     D    The rate the refractive index is changing at R
C
C Vax 11/750   1984 August               Catherine Hohenkerk
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DOUBLE PRECISION N,NT
C
      B = A/TT
      N = 1.0D0 + (NT - 1.0D0)*EXP(-B*(R-RT))
      DNDR = -B*(NT-1.0D0)*EXP(-B*(R-RT))
C
      END
```